Paper 238-2009

An analysis of market shares on the Danish alcohol market using unobserved components

Anders Milhøj, University of Copenhagen, Denmark

ABSTRACT

The Danish alcohol market has three types of alcohol: Beer, Wine and Spirits. The market share of wine has doubled over a 25 year period, while the market share of beer has been declining and the market share of spirits is generally low and fluctuating. In recent years the trending behavior has however changed, most likely because of changes in taxation on spirits. In the paper these market shares are analyzed by unobserved components models using Proc Ucm as models with time varying trends etc. are well suited for this type of data. In Denmark the relative prices for the three types of alcohol have changed radically because of changes in the taxation and hence the relative prices provide good independent variables in regressions. In SAS version 9.2 Proc Ucm has been extended with a Randomreg statement, that allows for the varying regression coefficients. One result is that the effect of relative prices among the three types of alcohol has decreased, probably because of the increasing wealth in Denmark.

INTRODUCTION

Models for unobserved components in time series analysis have been known for a long time. The idea is that a time series could be decomposed into a sum of a trend, seasonal factors and irregular components. These components each has its individual interpretation which seems appealing. This idea has been extended to a very flexible definition as all suggested components could in fact be time varying by an application of the Kalman filter and moreover explanatory variables with time varying coefficients could be included in the decomposition. This idea is used in seasonal adjustments as implemented in Proc X11 and Proc X12 in the SAS system. But for modeling purposes the SAS system also has the specific procedure, Proc Ucm, which directly points at the time series properties of the series in question. The procedure Proc Ucm could be seen as a user friendly access to Proc Statespace for Kalman filtering.

In this paper some of the new features of Proc Ucm in SAS version 9.2 are applied, but the theoretical properties of the underlying statistical model and the Kalman filtering are not in anyway stated or explicitly discussed. The two series being analyzed in this paper are the market shares for wine and spirits in the Danish alcohol market. The two time series behave differently as the market share of wine has an upward trending behavior while the market share of spirits is rather constant though fluctuating. Some of the fluctuations in the series are explained using the relative prices as regressors in a very flexible way using the possibility for adaptive modeling in Proc Ucm.

The Danish alcohol market

In this paper the market shares for beer, wine and spirits in the Denmark are considered using log transformed relative prices as explanatory variables. The log transformed market shares are modelled by unobserved components using Proc Ucm in SAS version 9.2 allowing for dynamic developments in the series and allowing for unstable relationships. The analysis is based on 29 observations for the years 1978 – 2006, while the observation for 2007 is used as a benchmark in order to study forecasting performance.

The total Danish sales of alcohol increased steadily for many years up to the late seventies, but the last say 25 years it has been remarkably constant. For many years the sales of wine have increased while the sales of beer and spirits have decreased, see Figure 1 where all series are plotted. Note that all sales data has been converted to 100% alcohol. Important reasons for this change in market shares are changes in eating habits towards more Mediterranean food like pasta but also the knowledge of wine, which is not a domestic product, has increased as many spend their holidays in southern Europe. Also price relations have changed as the Danish taxation on spirits some years ago was heavier than the taxation on wine which again was heavier than the taxation on beer. These different levels of taxation could be seen as a way of favouring domestic production – a point that was stressed by the

fact that the taxation on sprits was defined such that imported spirits were taxed more heavily than the traditional Danish snaps. The changes in taxation in recent years are all in the direction of taxation on alcohol content independent on the type of alcohol. This is the reason for the rather dramatic changes seen in Figure 2 which shows index series for the prices as they are met by the consumers.



Figure 1. The Danish sales of alcohol.



Figure 2. Price indices for beer, wine and spirits in Denmark.

Figure 3 shows the log transformed market share of wine and for spirits. The market share of wine has increased almost linearly for many years but the linear upward trend seems to end around 2002. The market share of spirits declined for many years but tends to be rising in recent years. This could be due to the latest regulation of the taxation which lowered the tax on spirits leaving the taxations on wine and beer unchanged.



Figure 3. Log-transformed market shares of wine and spirits.

When studying these market shares over a span of many years one should in theory take many changes in society into account such as above mentioned changes in eating and drinking habits, changes in income, which means that the Danes could afford to buy expensive spirits if they really wanted to do so and the increase in the population of Denmark which on the other hand includes many non drinking immigrants. Moreover the border trade with alcohol is important, as Danes could save by buying alcohol in Germany while Swedes and Norwegians could save by buying alcohol in Denmark. The net-effect of this border trade is hard to find precise data for and hence the present analysis consider the Danish alcohol market as simply the market for alcohol sold with a Danish taxation.

All these considerations lead to the conclusion that a formal statistical model with constant parameter values is impossible to formulate as important independent variables are not at hand and the dynamics possibly might have changed during the period under study. For these reasons models using the idea of unobserved components are appealing.

Analysis of the market share of wine using Proc Ucm in SAS version 9.2

In this section Proc Ucm is applied in an analysis of the log transformed market share of wine. In the first procedure call both a level and a slope are included in the model. The option back=1 is included in order to let 2007 be a true ex post benchmark for forecasting, as the information for 2007 is excluded from both the estimation and the forecasting calculations. The outlier statement provides an outlier test and the option checkbreak in the level statement gives a possibility of checking whether shifts of levels are present, which is not the case for this series.

```
ods graphics;
proc ucm data = sasts.alcoholmarket;
      model lwineratio;
      irregular plot=smooth;
      level plot=(filter smooth) print=smooth checkbreak ;
      slope plot=smooth;
      outlier ;
      estimate plot=(panel all) back=1;
      forecast back=1 lead=10 plot=forecasts alpha=0.1;
      id year align=middle interval=year;
run;
ods graphics off;
```

For the level component both a filtered and a smoothed plot are drawn, Figures 4 and 5. The filtered plot is defined using only past observations such that only observations up to time t are included in the calculation of the level at time t. In the smoothed version all observations are used leading to a smoother ex post plot. The plots look very similar but as the smoothed plot is based on more observations the confidence limits are narrower.

The level, see Figure 4, is steadily rising up to 2001 due to the slope component, which is almost constant giving an annual trend in the market share of wine. The level 0.03 for the slope corresponds to an approximately 3% increase in the market share for wine in Denmark. Since 2001 the level seems to be constant and the slope as seen in Figure 6 has reduced to a non significant value, leading to the conclusion that the upward trend ended around 2001. This possibility to allow time varying trend parameters is seen to provide us with very flexible models.



Figure 4. The smoothed level component.



Figure 5. The filtered level component.



Figure 6. The smoothed slope component.



Figure 7. The smoothed irregular component.

The irregular component, Figure 7, seems to be more volatile in the first years of the study showing shifting signs, but since about 1990 the irregular component has become more smooth. This is also seen at the forecast plot, Figure 8, where the fit is close – at least showing no systematic errors – apart form a systematic overshoot for 2002 – 2004. The largest forecast errors are found in the early years of the study and in fact the observation for 1989 is pointed out as an outlier in the text output with a p-value 3.4%. The forecast for 2007 seems reasonable and the actual observation is within the prediction limits.

The plot option in the estimate statement gives a panel plot with a histogram, a Gaussian probability plot for the residuals and autocorrelation plots, as seen in Figure 9, and moreover various cusum plots and plots summarizing Ljung-Box tests for autocorrelation (not shown in the present paper). None of these plots indicate any conflict with a hypothesis of normally distributed independent errors.



Figure 8. Actual and forecasted values.



Figure 9. Model fit diagnostics.

In the text output the reported error variances are rather small and not significantly different from zero when judged from the reported t-tests. In the text output the value of the components for the last observation, year 2006, are given and they are tested by a χ^2-test. Again the level is of course significantly different from zero but the slope is insignificant for 2006 but judged from the plots the trend was significant up to the millennium shift. The idea behind models with unobserved components is that the estimated level and slope after a few observations capture a new behaviour – in this situation with no or at least not so marked upward trend.

The forecasts are defined using the latest observed values of the level and the slope components and of course forecasting the irregular component by the value zero. The forecasts indicate a slightly upward trend but the confidence limits are broad for longer horizons even if the confidence level is specified at α = 10% instead of the more commonly applied default value α = 5%.

In the text output the level component is printed as requested by the print option in the level statement and in this way it could be saved using the ODS table if necessary, e.g. as input in models for other series. The forecasts are printed by default.

In the analysis also information of relative prices is available in form of relative prices for wine to the two other types of alcohol beer and spirits. These price ratios could be incorporated in Proc Ucm in two ways – as fixed coefficient regressors or as random coefficient regressors.

First fixed coefficient regressors are applied as in the following call of Proc Ucm. Here the estimated coefficient of the relative wine to beer price ratio is non significant at a 5% test level, but the price ratio of wine to spirits is just significant when tested against a one sided alternative at a 5% level. If the price ratio of wine to beer is skipped as an independent variable in the model the significance of the wine to spirits ratio is strengthened a little.

```
ods graphics;
proc ucm data = sasts.alcoholmarket;
      model lwineratio=lwinebeer_priceratio lwinespirits_priceratio ;
      irregular plot=smooth;
      level  plot=smooth;
      slope  plot=smooth;
      estimate plot=(panel all) back=1;
      forecast back=1 lead=10 plot=forecasts alpha=0.1;
         id year align=middle interval=year;
run;
ods graphics off;
```

Component	Parameter	Estimate	Approx Std Error	t Value	Approx Pr > |t|
Irregular	Error Variance	0.00041184	0.0002302	1.79	0.0736
Level	Error Variance	0.00033078	0.0003500	0.95	0.3446
Slope	Error Variance	0.00000728	0.00002520	0.29	0.7726
lwinebeer_priceratio	Coefficient	0.06380	0.23233	0.27	0.7836
lwinespirits_priceratio	Coefficient	-0.15883	0.09498	-1.67	0.0945

The possibility of random coefficient regressors, which is a new feature in SAS version 9.2, is applied in the following call of Proc Ucm. Here the price ratio of wine to beer is removed as it is insignificant and the model is still specified with both a level and a slope. The plot option in the randomreg statement gives a smoothed plot, Figure 10, of the time varying regression coefficient.

```
ods graphics;
proc ucm data = sasts.alcoholmarket;
      model lwineratio;
      irregular plot=smooth;
      level  plot=smooth;
      slope  plot=smooth;
      randomreg  lwinespirits_priceratio /plot=smooth;
      estimate plot=(panel all) back=1;
      forecast back=1 lead=10 plot=forecasts alpha=0.1;
         id year align=middle interval=year;
run;
ods graphics off;
```



Figure 10. Adaptive estimation of the regression coefficient to the wine-spirits price ratio.



Figure 11. Actual and forecasted values.

Final Estimates of the Free Parameters

Component	Parameter	Estimate	Approx Std Error	t Value	Approx Pr > |t|
Irregular	Error Variance	0.00062634	0.0002023	3.10	0.0020
Level	Error Variance	4.47161E-10	2.50521E-7	0.00	0.9986
Slope	Error Variance	3.81563E-12	2.78334E-9	0.00	0.9989
lwinespirits_priceratio	Error Variance	0.00769	0.0096082	0.80	0.4235

Significance Analysis of Components (Based on the Final State)

Component	DF	Chi-Square	Pr > ChiSq
Irregular	1	0.00	0.9770
Level	1	4432.04	<.0001
Slope	1	1544.95	<.0001
lwinespirits_priceratio	1	11.22	0.0008

Trend Information (Based on the Final State)

Name	Estimate	Standard Error
Level	-0.8622327	0.0129516
Slope	0.028165622	0.0007166

In this estimation the price ratio for wine to sprits is clearly significant for the market share of wine in recent years, while it of course was less significant 20 years ago as the price ratio as seen on Figure 2 was almost constant up to say 1990. The regression coefficient, that is the cross price elasticity of wine to spirits is in recent years stabilised around - 0.30, which is numerically larger than the value - 0.16 which was estimated by the fixed regression technique. The slope parameter is estimated as a constant 0.028 as its error variance is practically zero and hence the visible break of the upward trend for the market share of wine in 2001 is in this model solely referred to the change in the price ratio of wine to spirits caused by the change in taxation in 2001. In the forecasting period the straight upward linear trend is assumed to continue and the market share for wine is forecasted as a linear trend with very narrow confidence limits. In fact the confidence limits are much narrower in Figure 11 than in Figure 8 for the model using no price information.

Analysis of the market share of spirits using Proc Ucm in SAS version 9.2

The market share of spirits has developed differently from the market share of wine in the period of study. The market share has been declining for many years but not in form of a steady downward trend but in recent years the market share has increased somewhat but at present the level seems to stabilise.

A first call of Proc Ucm as in the start of the analysis of market share of wine in the previous section gives a nearly perfect fit of the observations by the level and slope components leaving the irregular component as nearly zero. This is possible by letting the variance of the error term in the level component be too large compared to the variance of the irregular component. The variances are estimated as 0.002 for the level error term and 9E-9 for the irregular component. In order to give a more smooth fit to the observed time series the variance of the error term in the level component is forced to attain a lower value as in the following call of Proc Ucm.

```
ods graphics;
proc ucm data = sasts.alcoholmarket;
      model lspiritsratio;
      irregular plot=smooth print=smooth;
      level plot=smooth var=0.001 noest checkbreak ;
      slope plot=smooth;
      outlier ;
      estimate plot=(panel all) back=1;
      forecast back=1 lead=10 plot=forecasts alpha=0.1;
      id year align=middle interval=year;
run;
ods graphics off;
```



Figure 12. Smoothed slope component.



Figure 13. Actual and forecasted values.

Here the slope, see Figure 12, is fluctuating around zero and even if it is significant for some years one could argue that no even time varying slope is present. In recent years the slope has been positive and for this reason the forecasts are calculated using a slight upward trend. The fit seems reasonable in the observation period.

In the next call of Proc Ucm the price ratios of wine to sprits and of beer to spirits are applied as independent variables in the model allowing for time varying regression coefficients using the Randomreg statement. Moreover the slope statement is excluded as the slope in the previous analysis seemed to be insignificant.

```
ods graphics;
proc ucm data = sasts.alcoholmarket;
      model lspiritsratio;
      irregular plot=smooth;
      level var=0.0001 noest plot=smooth;
      randomreg  lwinespirits_priceratio lbeerspirits_priceratio /plot=smooth;
      estimate plot=(panel all) back=1;
      forecast back=1 lead=10 plot=forecasts alpha=0.1;
         id year align=middle interval=year;
run;
ods graphics off;
```



Figure 14. Adaptive estimation of the regression coefficient to the wine spirits price ratio.



Figure 15. Adaptive estimation of the regression coefficient to the beer spirits price ratio.



Figure 16. Smoothed irregular component.



Figure 17. Actual and forecasted values.

Here the price ratio of beer to spirits is significant when judged from the smoothed plots, while the price ratio of wine to spirits is clearly insignificant. These results indicate that the choice among types of alcohol are made between beer

and spirits and not that much between wine and spirits. The coefficient, the cross price elasticity between beer and spirits, is estimated close to one, but in recent years the coefficient to the price ratio of beer to spirits has reduced as an indication that the tendency to drink the cheapest possible way has reduced – possibly because of increasing wealth.

The fit seems to be reasonable as the irregular component, see Figure 16, seems to vary in an unsystematic way. The plot of predicted and actual values, Figure 17, also gives a nice fit and the forecast limits are in fact very narrow compared to the similar plot in Figure 13 where the price ratios are not included in the model.

CONCLUSION

It is an empirical fact that the total alcohol consumption in Denmark has been constant for many years, but this does not in any way mean that the market situation is static. By an application of models for unobserved components it is demonstrated, that the market share of wine is increasing by approximately 2.8% pr. year. Moreover the observed reduction in the market share of wine since 2001 is completely caused by the changes in the taxation which were in favour of spirits at the expense of beer and wine. The upward trend is in fact unchanged in the adaptive estimations but the component due to the price ratio gave a pause in the upward trend.

Another result is that the market share of sprits is more dependent on price ratio of beer to spirits than the price ratio of wine to spirits. Another result is that all price ratios lose importance in the sense that the regression coefficients, which are in fact cross price elasticises, are numerically decreasing towards zero. These conclusions are drawn by an application of features in the newly released SAS version 9.2.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Anders Milhøj
Department of Economics, University of Copenhagen
Studiestræde 6
DK 1455 Copenhagen K

Work Phone:+45 35323265
E-mail: anders.milhoj@econ.ku.dk
Web: www.econ.ku.dk/milhoj